Dror Bar-Natan: Classes: 2012-13: Math 240 Algebra I: Tuesday September 18:

The Complex Numbers by Computer

Theorem. There is a field Comp that contains R and also contains an element i so that $i^2 = -1$.

Definitions

```
Comp /: Comp[a_, b_] + Comp[c_, d_] := Comp[a + c, b + d];
Comp /: Comp[a_, b_] * Comp[c_, d_] := Comp[a * c - b * d, a * d + b * c];
Comp[0] = Comp[0, 0];
Comp[1] = Comp[1, 0];
Comp /: Simplify[Comp[a_, b_]] := Comp[Simplify[a], Simplify[b]];
z1 = Comp[a1, b1];
z2 = Comp[a2, b2];
z3 = Comp[a3, b3];
```

"Comp" is a field.

The "F1" Properties

```
{z1 + z2, z2 + z1}
```

```
{Comp[a1 + a2, b1 + b2], Comp[a1 + a2, b1 + b2]}
```

```
{z1 * z2, z2 * z1}
```

```
{Comp[a1 a2 - b1 b2, a2 b1 + a1 b2], Comp[a1 a2 - b1 b2, a2 b1 + a1 b2]}
```

The "F2" Properties

```
{(z1 + z2) + z3, z1 + (z2 + z3)}
```

```
{Comp[a1 + a2 + a3, b1 + b2 + b3], Comp[a1 + a2 + a3, b1 + b2 + b3]}
```

```
f2 = {(z1 * z2) * z3, z1 * (z2 * z3)}
```

```
{Comp[a3 (a1 a2 - b1 b2) - (a2 b1 + a1 b2) b3, a3 (a2 b1 + a1 b2) + (a1 a2 - b1 b2) b3],
 Comp[-b1 (a3 b2 + a2 b3) + a1 (a2 a3 - b2 b3), a1 (a3 b2 + a2 b3) + b1 (a2 a3 - b2 b3)]}
```

```
Simplify[f2]
```

```
{Comp[a1 a2 a3 - a3 b1 b2 - a2 b1 b3 - a1 b2 b3, a2 a3 b1 + a1 a3 b2 + a1 a2 b3 - b1 b2 b3],
 Comp[a1 a2 a3 - a3 b1 b2 - a2 b1 b3 - a1 b2 b3, a2 a3 b1 + a1 a3 b2 + a1 a2 b3 - b1 b2 b3]}
```

The "F3" Properties

```
{z1, z1 + Comp[0]}
```

```
{Comp[a1, b1], Comp[a1, b1]}
```

```
{z1, z1 * Comp[1]}
```

```
{Comp[a1, b1], Comp[a1, b1]}
```

The "F4" Properties

```
{z1 + Comp[-a1, -b1], Comp[0]}
```

```
{Comp[0, 0], Comp[0, 0]}
```

```
f4 = {z1 * Comp[a1/(a1^2 + b1^2), -b1/(a1^2 + b1^2)], Comp[1]}
```

$$\left\{\text{Comp}\left[\frac{a1^2}{a1^2+b1^2}+\frac{b1^2}{a1^2+b1^2},\ 0\right],\ \text{Comp}[1,\ 0]\right\}$$

```
Simplify[f4]
```

```
{Comp[1, 0], Comp[1, 0]}
```

The "F5" Property

```
f5 = {z1 * (z2 + z3), z1 * z2 + z1 * z3}
```

```
{Comp[a1 (a2 + a3) - b1 (b2 + b3), (a2 + a3) b1 + a1 (b2 + b3)],
 Comp[a1 a2 + a1 a3 - b1 b2 - b1 b3, a2 b1 + a3 b1 + a1 b2 + a1 b3]}
```

```
Simplify[f5]
```

```
{Comp[a1 (a2 + a3) - b1 (b2 + b3), (a2 + a3) b1 + a1 (b2 + b3)],
 Comp[a1 (a2 + a3) - b1 (b2 + b3), a2 b1 + a3 b1 + a1 (b2 + b3)]}
```

"Comp" contains the real numbers.

```
Comp[a1, 0] + Comp[a2, 0]
```

```
Comp[a1 + a2, 0]
```

```
Comp[a1, 0] * Comp[a2, 0]
```

```
Comp[a1 a2, 0]
```

"Comp" contains a square root of -1.

```
i = Comp[0, 1];
```

```
i * i
```

```
Comp[-1, 0]
```

```
i * i + Comp[1]
```

```
Comp[0, 0]
```

The Polar Presentation of Comp

```
Pol[r_, θ_] := Comp[r * Cos[θ], r * Sin[θ]]
```

```
res1 = Pol[r1, θ1] * Pol[r2, θ2]
```

```
Comp[r1 r2 Cos[θ1] Cos[θ2] - r1 r2 Sin[θ1] Sin[θ2],
 r1 r2 Cos[θ2] Sin[θ1] + r1 r2 Cos[θ1] Sin[θ2]]
```

```
Simplify[res1]
```

```
Comp[r1 r2 Cos[θ1 + θ2], r1 r2 Sin[θ1 + θ2]]
```